/**
 * Copyright (c) 2010, iDA MediaFoundry
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *
 * - Redistributions of source code must retain the above copyright notice,
 *   this list of conditions and the following disclaimer.
 *
 * - Redistributions in binary form must reproduce the above copyright notice,
 *   this list of conditions and the following disclaimer in the documentation
 *   and/or other materials provided with the distribution.
 *
 * - Neither the name of iDA MediaFoundry nor the names of its contributors may
 *   be used to endorse or promote products derived from this software without
 *   specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 */
package be.idamf.flash.exec
{
	/**
	 * Defers method execution to the next frame.
	 * 
	 * @author Jan Van Coppenolle
	 * 
	 * @param target 
	 */
	public function deferExec(target:Object, functionReference:Function, singleExec:Boolean = true, ... args):void
	{
		new DeferredExecution(target, functionReference, singleExec, args);
	}
}

import flash.display.Sprite;
import flash.events.Event;
import flash.utils.Dictionary;

internal class DeferredExecution
{
	private static const EXECUTIONER:Sprite = new Sprite();
	private static const REFERENCED:Dictionary = new Dictionary();

	public var target:Object;
	public var functionReference:Function;
	public var singleExec:Boolean;
	public var args:Array;

	public function DeferredExecution(target:Object, functionReference:Function, singleExec:Boolean, args:Array)
	{
		if (singleExec && REFERENCED[functionReference])
			return;

		this.target = target;
		this.functionReference = functionReference;
		this.singleExec = singleExec;
		this.args = args;

		EXECUTIONER.addEventListener(Event.ENTER_FRAME, executionHandler);

		if (singleExec)
			REFERENCED[functionReference] = true;
	}

	public function executionHandler(event:Event):void
	{
		EXECUTIONER.removeEventListener(Event.ENTER_FRAME, executionHandler);

		if (singleExec)
			delete REFERENCED[functionReference];

		functionReference.apply(target, args);
	}
}